START:


This is not for beginners and does not explain basic things. I recommend that you start off with some other available tutorial if your brand spanking new to (DIV) programming.


Now lets get set up. Start DIV up and open a new program and call it platform and save it in the default prg folder. Put these lines of code in it:

               PROGRAM PLATFORM;

               GLOBAL





 LOCAL


 BEGIN


    SET_MODE(m640x480);

                   FRAME;

                END

Next we want a ".fpg" folder. Go to the files menu open new save as "platform" in fpg folder. Now the first graphic we want to add is the player graphic for our purposes this graphic will be 34pixels high and 32 wide. With its centre set to (16,32).

   To change the centre open up the graphic for editing open up the control points bar and click on (16,32).

Save this as graphic 10 in the fpg and call it player.

   We want a process for our player so we change our code to now look like this:

CODE CHECK:

PROGRAM PLATFORM;

    GLOBAL




    LOCAL

    BEGIN

    SET_MODE(m640x480);

    PLAYER(100,30);

    FRAME;

    END

PROCESS PLAYER(X,Y);

PRIVATE

BEGIN

  Graph=10;

LOOP

 FRAME;

END

END

Before starting code for movement we need two more maps, one for the background so we can see where our player is and one for collision i.e. a hardness map. The background map will be 640x480 pixels 

Here is a simple version :

The hardness map is a greyscale version of the background map at half the size like so:

Save these maps as codes 1 and 2 in the platform ".fpg" respectively.


Now to add movement to our player we will start with horizontal movement. We want the player to move to the left at a certain speed if  the left key is pressed and stop if a wall is hit and vice-versa for when the right key is pressed.

       New variables:

               Local

                   speed;         //Speed at which to move player

and in the player process:

        incx;               //Both of these are for checking collision with  a wall and movement

        last_speed;   

CODE:

    IF(key(_right) and speed<8)

            speed+=2;

             flags=0; 

    ELSE

         IF(key(_left) and speed>-8)

            speed-=2;

             flags=1;                      //flips graphic horizontally

         ELSE

              IF(NOT key(_left) AND NOT key(_right));

                  IF(speed>0)

                       speed--;


    END


    IF(speed<0)


        speed++;



    END

              END

          END

     END

EXPLANATION OF CODE:

       Speed is the distance we will move the character, if the right key is pressed the distance increases until the limit is met (8). If neither the left or right key is pressed the speed will reduce itself to zero. If the speed is less then zero the player will move to the left if it's greater the zero the player will move to the right.

CODE FOR MOVEMENT:

last_speed=speed;

IF(last_speed<>0)      //last speed not equal to zero

    incx=last_speed/abs(last_speed);     //this is the movement variable= either plus or minus one

    WHILE(last_speed<>0)

       x+=incx;  

       last_speed -=incx;

    END

END

EXPLANATION:

     Assume speed is equal to minus ten and x=0 =>

        Last_speed=speed= minus ten;

         Last speed is not equal to zero   =>    

            incx=last_speed/abs(last_speed)= minus ten divide by ten equal minus one

            while last_speed isnt equal to zero

                  x=incx;//increment x => x=minus one

                  last_speed -= incx = (-10)-(-1)=-9

//keep going till last_speed equals 0      

             end  

 So this bit of code moves the player the distance required.

CODE CHECK:

PROGRAM platform;

GLOBAL
LOCAL
 SPEED;

BEGIN

 load_fpg("fpg\platform.fpg");

 set_mode(m640x480);

 put_screen(0,1);

 player(100,30);

END
PROCESS player(x,y)

PRIVATE
 incx;

 last_speed;

BEGIN
 graph=10;

 LOOP
  // Reads keyboard

  IF (key(_right)  AND SPEED<8)

    SPEED+=2;           // Increases speed to the right

    flags=0;
  

  ELSE

   IF (key(_left) AND SPEED>-8)

     SPEED-=2;       // Increases speed to the left

     flags=1;
   

   ELSE                // If no key is pressed

     IF(not key(_LEFT) and not key(_RIGHT))

      IFSPEED>0)    // Brakes to the right

       SPEED--;

      END

      IF (SPEED<0)    // Or brakes to the left

        SPEED++;

      END

     END

   END

  END

  last_speed=speed;

  IF(last_speed<>0)

   incx=last_speed/ABS(last_speed);

   WHILE(last_speed<>0)

    x+=incx;    // And if so, moves character

    last_speed-=incx;

   END

  END

  frame;

 END //END OF LOOP

END

SEE example1.prg for source.

Now what we have is a process that moves from left to right but doesn’t stop for walls, lets change that. First up open up the hardness map from our graphics file for editing. Now select a colour from the palette to represent a wall. Make sure the colour is not close too either back, white or any greys pick a nice bright colour. My fpg uses the normal div palette and I chose colour 73 from the palette a nice yellow colour. Now colour all your walls as the picture below you might want to do a double thickness.

Save this map back into the fpg at the same number. Now we need to add code to check for this colour

in our player process.

CODE:

      IF(map_get_pixel(0,2,(x+incx)/2,(y-2)/2)<>73)

           IF(map_get_pixel(0,2,(x+incx)/2,(y-32)/2)<>73)


    x+=incx;

           END  

       END 

EXPLANATION:

    What this code does is check the next pixel beside the process and if this pixel is not yellow x increases if the pixel colour is yellow the process stays where it is. The first if statement checks 2pixels above where the bottom of the graphic is for a wall and if there is no wall there it checks halfway up the graphic for a wall and if there is no wall there it moves the process.

The reason the x and y values are divided by two is that the hardness map is half the size of the background map and therefore these are the corresponding points to the screen. This code goes in the players while loop.

CODE CHECK:

PROGRAM platform;

GLOBAL
LOCAL
 SPEED;

BEGIN
 load_fpg("fpg\platform.fpg");

 set_mode(m640x480);

 put_screen(0,1);

 player(100,30);

END
PROCESS player(x,y)

PRIVATE
 incx;

 last_speed;

BEGIN

 graph=10;

 LOOP

  // Reads keyboard

  IF(key(_right)  AND SPEED<8)

    SPEED+=2;           // Increases speed to the right

    flags=1;

  ELSE

   IF(key(_left) AND SPEED>-8)

     SPEED-=2;       // Increases speed to the left

     flags=0;
   

   ELSE                // If no key is pressed

     IF(not key(_LEFT) and not key(_RIGHT))

      IF(SPEED>0)    // Brakes to the right

       SPEED--;

      END

      IF(SPEED<0)    // Or brakes to the left

        SPEED++;

      END

     END

   END

  END
  last_speed=speed;

  IF(last_speed<>0)

   incx=last_speed/ABS(last_speed);

   WHILE(last_speed<>0)

            IF(map_get_pixel(0,2,(x+incx)/2,(y-2)/2)<>73)              //NEW CODE

              IF(map_get_pixel(0,2,(x+incx)/2,(y-32)/2)<>73)

    x+=incx;

              END  

             END 

    last_speed-=incx;

   END
  END
 frame;

 END //END OF LOOP

END                  //SEE example2.prg for source code 
Run your program now. What it has now is a character moving from left to right and stopping when it collides with walls. We're nearly done there's just one more thing to do and that’s add some gravity.

 First things first open up our hardness map again and pick a colour as before(make sure its not too close to the first colour) and colour in the floors as below I used colour 22 for my floors a nice red colour(although it looks orange in reality its red ;-).

You need to go over this colour in certain places make sure you have a double thickness for flat floors. If the floor is very complicated put

More lines down. For example for the little slope I have I put three lines of red down.

Now onto the gravity part. We will use a separate process for the gravity it doesn’t need to be done in a separate process but if you want to give other process` gravity its easier to do then writing all the code out again in each process. The process doesn't need a frame statement as it will use its father's frame. There is only one private variable and that is vgravity which is a temporary storage variable.

The implementation is quite like the horizontal movement except here it works out itself how far, what speed and when to move. We have one new local variable and that is gravity_speed. If the gravity speed is greater the zero(positive) our player will move down if it is less the zero our player will move up and if its equal to zero our player will be on the ground. One more new variable and that’s also a local variable and this will be on_the_ground this will tell us when the process is on the ground. Lets start then.

CODE:

  In the player process put this

    IF(key(_control) and on_the_ground==true)

  
On_the_ground=false;   

      gravity_speed=-18; //the lower this is the higher the jump

    END

  and put in a call to the gravity process

   gravity();

  and create a new process gravity.

   PROCESS gravity();

   PRIVATE

    vgravity;

   BEGIN

   END

Now to flesh the process out. We need to fill our temporary variable vgravity so


vgravity= (father.gravity_speed+=2);

now we want a limit on how fast the character can fall so


IF(vgravity>16)

        vgravity=16;


  father.gravity_speed=16;

      END

now if gravity is less the zero move the character up


IF(vgravity<0)

       WHILE(vgravity++!=0)  //This adds one to the vgravity as well

         father.y--;

       END

other wise if vgravity is greater then zero move character down

    
ELSE


  father.y+=vgravity

check for collision with the floor

 FROM vgravity=-16 TO 7 STEP 2;     // Ends process if floor is touched

   IF (map_get_pixel(1,2,father.x/2,(father.y+vgravity)/2)==22)

     BREAK;                      // Exit if floor is touched

   END

 END

 IF (vgravity<8)                    // If floor is collided

   father.y+=vgravity;    

   father.gravity_speed=0;        // And gravity is set to 0

   father.on_the_ground=true;  

 END

END //IF ELSE

I will explain this last part(the ELSE statement) but not the previous as I think you should understand that.

So first of all, the father process is moved down (father.y+=vgravity). Then the FROM statement checks to see if the process collides with the floor it checks 16 pixels above the graphics centre(16,32) and 7 pixels below it. If on any of these checks the pixel is red(colour number 22) the floor has been hit and the FROM statement is ended by a BREAK and vgravity is less then eight. If on the other hand all these pixels are checked and none of them equal red then the floor hasn’t been hit and the process continues falling.

CODE CHECK:

PROGRAM platform;

GLOBAL
LOCAL
 SPEED;

 on_the_ground;

 gravity_speed;

BEGIN
 load_fpg("fpg\platform.fpg");

 set_mode(m640x480);

 put_screen(0,1);

 player(100,30);

END
PROCESS player(x,y)

PRIVATE
 incx;

 last_speed;

BEGIN

 graph=10;

 LOOP

  // Reads keyboard

  IF (key(_right)  AND SPEED<8)

    SPEED+=2;           // Increases speed to the right

    flags=0;


  ELSE

   IF (key(_left) AND SPEED>-8)

     flags=1;

     SPEED-=2;       // Increases speed to the left

   ELSE                // If no key is pressed

     IF(not key(_LEFT) AND not key(_RIGHT))

      IF (SPEED>0)    // Brakes to the right

       SPEED--;

      END

      IF (SPEED<0)    // Or brakes to the left

        SPEED++;

      END

     END

   END

  END
  last_speed=speed;

  IF(last_speed<>0)

   incx=last_speed/ABS(last_speed);

   WHILE(last_speed<>0)

            IF(map_get_pixel(0,2,(x+incx)/2,(y-2)/2)<>73)              

              IF(map_get_pixel(0,2,(x+incx)/2,(y-32)/2)<>73)


    x+=incx;

              END  

             END 

    last_speed-=incx;

   END
  END
  IF(KEY(_CONTROL) AND on_the_ground==true)

    on_the_ground=false;

    gravity_speed=-18;

  END

  gravity();  

 frame;

 END //END OF LOOP

END
PROCESS gravity();

PRIVATE

 vgravity;

BEGIN

 vgravity=(father.gravity_speed+=2);

 IF(vgravity>16)        

   vgravity=16;  

   father.gravity_speed=16; 

 END

 IF(vgravity<0)                        // If gravity is negative

   WHILE (vgravity++!=0)      // While not zero, increments it

    IF(map_get_pixel(0,2,father.x/2,(father.y-24)/2)<>252)

      father.y--;  

    END

   END

 ELSE   // If gravity is zero or positive

   father.y+=vgravity; 

   FROM vgravity=-16 TO 7 STEP 2;   

    IF(map_get_pixel(0,2,father.x/2,(father.y+vgravity)/2)==22)

      BREAK; 

    END

   END

   IF (vgravity<8)                    

     father.y+=vgravity;  

     father.gravity_speed=0;  

     father.on_the_ground=true;   

   END

 END  

END//END OF PROCESS

SEE example3.prg for source code.

